
Diese Beschreibung gilt für den TI Voyage 200;  TI 92 Plus, TI 92 und TI 89 arbeiten mit nahezu denselben Eingaben. Das Programm ist auf einem Voyage 200 getestet.

Einstellung des Rechners (über MODE-Taste): 'approximate' und 'float 12' Genauigkeit.

Die Funktion bode2f  liefert für eine vor dem Aufruf zu definierende Funktion f(x,y) eine Näherung des Integralwerts auf einem Funktionsstreifenbereich



  


B = { (x,y) │a≤x≤b, c(x)≤y≤d(x) }
nach dem 2-dim. BODE-Verfahren für Funktionsstreifen (vgl. Vorlesung Numerik).

Die Grenzfunktionen müssen auf [a,b] die Bedingung c(x)≤d(x) erfüllen und werden ebenfalls vor dem Funktionsaufruf eingegeben. Ein Aufruf-Beispiel:

0 → c(x)



© Definition der unteren Grenzfunktion   (hier c(x)=0=konst. )
ln(x) → d(x)


© Definition der oberen Grenzfunktion    (hier d(x)=ln(x) )
x*e^y → f(x,y)
   
© Definition der zu integrierenden Funktion f(x,y)  (zB f(x,y)= xey)
Die Reihenfolge dieser drei Funktionseingaben spielt keine Rolle.

Schließlich wird die Funktion bode2f  wie folgt aufgerufen:

bode2f(a,b,n, m)

©  a,b sind die Intervallgrenzen für x,





©  n ist die Zerlegungszahl für x, m für y (beide durch 4 teilbar!)
Zur Logik vergleiche man die Programme bode, simpson2 und simp2f.

Ein Testlauf mit der obigen Beispielfunktion  f(x,y) = xey  liefert
für den Integrationsbereich  B = {(x,y) │1≤x≤2, 0≤y≤ln(x) }
mit den Zerlegungszahlen n=4 für x und m=4 für y

durch den Aufruf

0 → c(x)



© untere Grenzfunktion





ln(x) → d(x)


© obere Grenzfunktion





x*e^y → f(x,y)

© zu integrierende Funktion



 

bode2f(1,2,4, 4)

© Aufruf der 2dim BODE-Funktion

den Wert B4x8 = 0.833 333 349 675 (exakt ist 0.8333333....).

Das Beispiel zeigt, dass auch konstante Grenzfunktionen eingesetzt werden können.

Der Programm-Code folgt auf der nächsten Seite.

Ein zweiter Testlauf mit  f(x,y) = exp(√(x+y)) auf  B = {(x,y) │0≤x≤1, x2≤y≤ex }

liefert mit n=8 und m=8 mit folgenden Eingaben






x^2 → c(x)






e^x → d(x)






exp(√(x+y)) → f(x,y)






bode2f(0,1,8,8)

den Integralwert 4.994 850 338 69.
Programmcode bode2f
bode2f(a,b,n, m)
Func

Local h,ki,p,q,i,j,x,y,si,s
(b-a)/n→h





© Schrittweite h für x

newList(n+1)→p




© p und q sind Vektoren, die die BODE-Koeff.
newList(m+1)→q




© enthalten: [7,32,12,32,14,...,14,32,12,32,7]








© da List-Variable den Index 0 nicht zulassen, gilt









© p[1]=7, p[2]=32, p[3]=12 usw (1 + normaler Index)

7→p[1] : 7→p[1+n]



© Aufbau der BODE-Gewichte für x im Vektor p

For i,1,n-1


If mod(i,2)=0 Then


If mod(i,4)=0 Then




14→p[1+i]



Else




12→p[1+i]



EndIf


Else



32→p[1+i]


EndIf

EndFor

7→q[1] : 7→q[1+m]



© Aufbau der BODE-Gewichte für y im Vektor q
For j,1,m-1


If mod(j,2)=0 Then


If mod(j,4)=0 Then




14→q[1+j]



Else




12→q[1+j]



EndIf


Else



32→q[1+j]


EndIf

EndFor

0→s






© Doppelsummation nach BODE2
For i,0,n

a+i*h→x : 0→si

(d(x) - c(x)) / m → ki


© Schrittweite ki für y abhängig von x bzw i


For j,0,m



c(x)+j*ki→y



si + q[1+j]*f(x,y) → si


EndFor


s+ p[1+i]*si*ki*2/45 → s

EndFor

return s*h*2/45
EndFunc

